```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the system of differential time-delay equations defining the wind tunnel model studied in A. Manitius, "Feedback controllers for a wind tunnel model involving a delay: analytical design and numerical simulations", *IEEE Trans. Autom. Contr.*, 29 (1984), 1058-1068. The system matrix is defined by

```
> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[t,s],
> comm=[a,k,omega,xi]):
```

```
> R:=matrix(3,4,[d+a,k*a*delta,0,0,0,d,-1,0,0,omega^2,d+2*xi*omega,-omega^2]);
```

$$R := \begin{bmatrix} d+a & k\,a\,\delta & 0 & 0 \\ 0 & d & -1 & 0 \\ 0 & \omega^2 & d+2\,\zeta\,\omega & -\omega^2 \end{bmatrix}$$

where a, k, ω and ζ are real parameters of the system. We introduce the $A = \mathbb{Q}(a,k,\omega,\zeta)[d,\delta]$-module $M = A^{1\times 4}/(A^{1\times 3}\,R)$. Let us compute the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A,mult_table);
```

$$Endo := [[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & -k\,a\,\delta & 0 & 0 \\ 0 & a & 1 & 0 \\ 0 & -\omega^2 & a-2\,\zeta\,\omega & \omega^2 \\ 0 & 0 & 0 & d+a \end{bmatrix}], \begin{bmatrix} d+a & -1 \end{bmatrix},$$
$$\begin{bmatrix} [1,1] & 1 & 0 \\ [1,2] & 0 & 1 \\ [2,1] & 0 & 1 \\ [2,2] & 0 & d+a \end{bmatrix}]$$

Hence, the A-module structure of E is defined by two generators id_M and f_1 defined by $f_1(\pi(\lambda)) = \pi(\lambda\,P_1)$, where $\pi : A^{1\times 4} \longrightarrow M$ denotes the canonical projection onto M, $\lambda \in A^{1\times 4}$, and P_1 is the second matrix of $Endo[1]$. The second matrix $Endo[2]$ of $Endo$ corresponds to the relation between the two generators $\{\mathrm{id}_M, f_1\}$ of E, i.e., we have $f_1 = (d+a)\,\mathrm{id}_M$. Hence, we obtain that E is a free A-module of rank 1 generated by id_M. The matrix formed by $Endo[3]$ but the first column is the trivial multiplication of the generators id_M and f_1 of E, namely:

$$\mathrm{id}_M \circ \mathrm{id}_M = \mathrm{id}_M, \quad \mathrm{id}_M \circ f_1 = f_1 \circ \mathrm{id}_M = f_1, \quad f_1 \circ f_1 = (d+a)\,f_1.$$

As the A-module E is generated by id_M, we obtain that an endomorphism f of M has the form $f = \alpha\,\mathrm{id}_M$, with $\alpha \in A$. Hence, the relation $f^2 = f$ implies that $\alpha^2\,\mathrm{id}_M = \alpha\,\mathrm{id}_M$, and thus, $\alpha = 0$ or $\alpha = 1$, i.e., $f = 0$ and $f = \mathrm{id}_M$ are the only two idempotents of E. In particular, we deduce that the A-module M is irreducible (see, e.g., Corollary 3.1 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381).

In particular, let us check that E does not have non-trivial constant idempotents:

```
> Idem:=IdempotentsConstCoeff(R,Endo[1],A,0,alpha);
```

$$Idem := [[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}], [Ore\_algebra, [``diff'', dual\_shift],$$

$$[t, s], [d, \delta], [t, s], [a, k, \omega, \zeta], 0, [], [], [t, s], [], [], [diff = [d, t], dual\_shift = [\delta, s]]]]$$

Even if the only two idempotent endomorphisms of M are the trivial ones, namely, id_M and 0, we can search for homotopies of id_M or 0 which allow us to find a block-diagonal matrix equivalent to R with a block equals to I_m. If so, then we can reduce the number of equations defining the differential time-delay linear system $\ker_{\mathcal{F}}(R.)$, where $\mathcal{F}$ denotes an A-module (e.g., $\mathcal{F} = C^{\infty}(\mathbb{R})$). Let us first denote by $P_1 = I_4$, $Q_1 = I_3$ and $Z_1 = 0$:

```
> P[1]:=Endo[1,1]; Q[1]:=diag(1$3); Z[1]:=Factorize(diag(0$3),R,A);
```

$$P_1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad Q_1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad Z_1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Let us compute the constant solutions of the algebraic Riccati equation $\Lambda\, R\, \Lambda + \Lambda = 0$ (for more details, see Section 4 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381):

```
> Mu:=RiccatiConstCoeff(R,P[1],Q[1],Z[1],A,0,alpha);
```

$$Mu :=$$

$$\left[\left[\begin{array}{ccc} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{array}\right], \left[\begin{array}{ccc} -b331 & \frac{(b311\, ab331 - 2\, b331\, b311\, \zeta\, \omega + b331\, \omega^2 b411 - b311)b331}{b311^2} & -\frac{b331^2}{b311} \\ 0 & 0 & 0 \\ b311 & -\frac{b311\, ab331 - 2\, b331\, b311\, \zeta\, \omega + b331\, \omega^2 b411 - b311}{b311} & b331 \\ b411 & -\frac{b411\,(b311\, ab331 - 2\, b331\, b311\, \zeta\, \omega + b331\, \omega^2 b411 - b311)}{b311^2} & \frac{b411\, b331}{b311} \end{array}\right],\right.$$

$$\left[\begin{array}{ccc} 0 & b131\, \omega^2 b421 & b131 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & b421 & \omega^{-2} \end{array}\right], \left[\begin{array}{ccc} 0 & 0 & b131 \\ 0 & 0 & \frac{-1+\omega^2 b431}{\omega^2} \\ 0 & 0 & 0 \\ 0 & 0 & b431 \end{array}\right], \left[\begin{array}{ccc} 0 & b121 & 0 \\ 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & b421 & 0 \end{array}\right],$$

$$\left[\begin{array}{ccc} 0 & -\frac{b131\, b331}{b231} & b131 \\ 0 & -b331 & b231 \\ 0 & -\frac{b331^2}{b231} & b331 \\ 0 & -\frac{(b331^2 + 2\, b231\, b331\, \zeta\, \omega + \omega^2 b231^2 + b231)b331}{\omega^2 b231^2} & \frac{b331^2 + 2\, b231\, b331\, \zeta\, \omega + \omega^2 b231^2 + b231}{\omega^2 b231} \end{array}\right],$$

$$\left.\left[\begin{array}{ccc} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ b411 & b421 & \omega^{-2} \end{array}\right], \left[\begin{array}{ccc} 0 & -\frac{1}{b411\, \omega^2} & 0 \\ 0 & 0 & 0 \\ 0 & 1 & 0 \\ b411 & -\frac{a - 2\,\zeta\,\omega}{\omega^2} & \omega^{-2} \end{array}\right]\right],$$

$$[Ore\_algebra, [\text{“}diff\text{''}, dual\_shift], [t, s], [d, \delta], [t, s], [a, k, \omega, \zeta, b231, b431, b421, b411,$$
$$b331, b311, b131, b121], 0, [], [], [t, s], [], [], [diff = [d, t], dual\_shift = [\delta, s]]]]$$

We find 8 constant solutions of the previous algebraic Riccati equation. Let us take the last one where we set the arbitrary constant $b411$ to 1:

```
>  Lambda:=subs(b411=1,Mu[1,8]);
```

$$\Lambda := \left[\begin{array}{ccc} 0 & -\omega^{-2} & 0 \\ 0 & 0 & 0 \\ 0 & 1 & 0 \\ 1 & -\frac{a - 2\,\zeta\,\omega}{\omega^2} & \omega^{-2} \end{array}\right]$$

We can consider the homotopy of id_M defined by the pair of matrices $P_2 = P_1 + \Lambda\, R$ and $Q_2 = Q_1 + R\, \Lambda$ defined by:

```
>  P[2]:=simplify(evalm(P[1]+Mult(Lambda,R,A)));
>  Q[2]:=simplify(evalm(Q[1]+Mult(R,Lambda,A)));
```

$$P_2 := \left[\begin{array}{cccc} 1 & -\frac{d}{\omega^2} & \omega^{-2} & 0 \\ 0 & 1 & 0 & 0 \\ 0 & d & 0 & 0 \\ d + a & -\frac{-\omega^2 - k\, a\, \delta\, \omega^2 + d\, a - 2\, d\, \zeta\, \omega}{\omega^2} & \frac{d+a}{\omega^2} & 0 \end{array}\right] \quad Q_2 := \left[\begin{array}{ccc} 1 & -\frac{d+a}{\omega^2} & 0 \\ 0 & 0 & 0 \\ -\omega^2 & d + a & 0 \end{array}\right]$$

We can now check that we have $R\, P_2 = Q_2\, R$, $P_2^2 = P_2$ and $Q_2^2 = Q_2$:

```
> VERIF1:=simplify(evalm(Mult(R,P[2],A)-Mult(Q[2],R,A)));
> VERIF2:=simplify(evalm(Mult(P[2],P[2],A)-P[2]));
> VERIF3:=simplify(evalm(Mult(Q[2],Q[2],A)-Q[2]));
```

$$VERIF1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad VERIF3 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

In particular, the endomorphism $e \in E$ defined by $e(\pi(\lambda)) = \pi(\lambda\, P_1)$, where $\lambda \in A^{1\times 4}$, satisfies $e^2 = e$, i.e., defines an idempotent of E. As e was obtained from id_M by means of a homotopy, we have $e = \mathrm{id}_M$. However, as we have $P_2^2 = P_2$ and $Q_2^2 = Q_2$, we know that the A-modules $\ker_A(.P_2)$, $\mathrm{im}_A(.P_2) = \ker_A(.(I_4 - P_2))$, $\ker_A(.Q_2)$ and $\mathrm{im}_A(.Q_2) = \ker_A(.(I_3 - Q_2))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of those free A-modules:

```
> U1:=SyzygyModule(P[2],A): U2:=SyzygyModule(evalm(1-P[2]),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[2],A); V2:=SyzygyModule(evalm(1-Q[2]),A):
> V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} d\,\omega^2 + \omega^2\,a & \omega^2 + k\,a\,\delta\,\omega^2 & d + 2\,\zeta\,\omega & -\omega^2 \\ 0 & d & -1 & 0 \\ \omega^2 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix} \quad V := \begin{bmatrix} \omega^2 & 0 & 1 \\ 0 & 1 & 0 \\ -\omega^2 & d + a & 0 \end{bmatrix}$$

The matrices $U \in \mathrm{GL}_4(A)$ and $V \in \mathrm{GL}_3(A)$ are such that $U\,P_2\,U^{-1}$ and $V\,Q_2\,V^{-1}$ are two block-diagonal matrices formed by the diagonal matrices 0_n and I_m:

```
> VERIF1:=Mult(U,P[2],LeftInverse(U,A),A);
> VERIF2:=Mult(V,Q[2],LeftInverse(V,A),A);
```

$$VERIF1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Hence, the matrix R is equivalent to the block-diagonal matrix $S = V\,R\,U^{-1}$ defined by:

```
> S:=Mult(V,R,LeftInverse(U,A),A);
```

$$S := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -d - a & d^2 - k\,a\,\delta\,\omega^2 + d\,a \end{bmatrix}$$

Hence, we have $M \cong A^{1\times 2}/(A\,(-(d+a) \quad d^2 + a\,d - k\,a\,\omega^2\,\delta))$. This result can be obtained by means of the command *HeuristicDecomposition*:

```
> HeuristicDecomposition(R,P[1],A)[1];
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -d - a & d^2 - k\,a\,\delta\,\omega^2 + d\,a \end{bmatrix}$$

We note that we can simplify again the last row of S by means of elementary column operations:

```
> X:=diag(diag(1$2),evalm([[-1,d],[0,1]]));
```

$$X := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & d \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

The new matrix $S\,X$ has then the simple form:

```
> Mult(S,X,A);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & d+a & -k\,a\,\delta\,\omega^2 \end{bmatrix}$$

Hence, if we denote by $Y = X^{-1}\,U \in \mathrm{GL}_4(A)$

```
> Y:=Mult(LeftInverse(X,A),U,A);
```

$$Y := \begin{bmatrix} d\,\omega^2 + \omega^2\,a & \omega^2 + k\,a\,\delta\,\omega^2 & d + 2\,\zeta\,\omega & -\omega^2 \\ 0 & d & -1 & 0 \\ -\omega^2 & d & -1 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

then we obtain that R is equivalent to the block-diagonal matrix $T = V\,R\,Y^{-1}$ defined by:

```
> T:=Mult(V,R,LeftInverse(Y,A),A);
```

$$T := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & d+a & -k\,a\,\delta\,\omega^2 \end{bmatrix}$$

If $\mathcal{F}$ denotes an A-module (e.g., $\mathcal{F} = C^\infty(\mathbb{R})$), then the linear differential time-delay system $\ker_{\mathcal{F}}(R.)$ is equivalent to the linear system $\ker_{\mathcal{F}}(S.)$ defined by the sole first order equation:

$$\dot{y}(t) + a\,y(t) - k\,a\,\omega^2\,v(t-h) = 0.$$